AD-A087 447 MASSACHUSETTS INST OF TECH LEXINGTON LINCOLN LAB F/G 4/1
CAD: A CODE FOR THE CALCULATION OF ATMOSPHERIC DENSITY.(U)
APR 80 W L OLIVER F19628-80-C-0002
UNCLASSIFIED TN-1980-20 ESD-TR-80-28 NL

1 of 1

END
DATE
FILMED
9-80
DTIC



MICROCOPY RESOLUTION TEST CHART


2

MASSACHUSETTS INSTITUTE OF TECHNOLOGY

LINCOLN LABORATORY

CAD: A CODE FOR THE CALCULATION OF ATMOSPHERIC DENSITY

W. L. OLIVER

Group 91

TECHNICAL NOTE 1980-20

2 APRIL 1980

Approved for public release; distribution unlimited.

LEXINGTON MASSACHUSETTS

ABSTRACT

An efficient machine implementation of the 1972 COSPAR International Reference Atmosphere (CIRA) is developed for use in evaluating atmospheric-drag effects upon earth-orbiting satellites. Restriction of this procedure to the satellite-orbit region of significant drag and the use of an alternate temperature profile representation have enabled a factor of 30 improvement in machine execution time compared with that required by the code supplied by CIRA publication while incurring only a few percent difference in the density results.

CONTENTS

I. INTRODUCTION

This note documents a computer code entitled CAD for the Calculation of Atmospheric Density at satellite altitudes. This code has been developed specifically for use by the ANODE (Analytic Orbit Determination) Program (Ref. 1), which is currently used at Millstone Hill as a rapid orbit estimator, to enable estimation of atmospheric drag effects upon satellite orbits, though the uses of this code could be more general.

CAD is intended to provide an efficient machine implementation of the CIRA (Cospar International Reference Atmosphere) 1972 (Ref. 2) density model. The CIRA publication provides a computer code entitled ADEN for this computation, but this code was "intended primarily as a reference, rather than as a working routine, ... in order to make the subroutine as simple and as closely related to the abstract description as possible. ... The subroutine, therefore, runs relatively slowly - probably too slowly for extensive use ... one call ... requires the equivalent of about 34000 floating point multiplications" (Ref. 2). Most of this calculation time is expended in the numerical integration of the diffusion equation from a lower boundary to the satellite altitude. CAD replaces this numerical integration with an analytic approximation, and it is shown that the difference between the two results is probably insubstantial in comparison with the uncertainty in such atmospheric models.

Reference 1 defines the "significant drag regime" of the atmosphere as being below 500 km altitude. The CIRA considers hydrogen to contribute negligibly to the total atmospheric density below this altitude, and thus the portion of the CIRA dealing with hydrogen may be omitted for the present purposes. Similarly, if we adopt 125 km as the lower boundary of interest for satellite orbits, a not unreasonable choice, then we may avoid the relatively complex portion of the CIRA applicable to altitudes below this level. In addition, the constituent argon, though considered in the CIRA, never contributes appreciably to total density, and this portion of the CIRA may also be omitted.

Included in the appendix is the FORTRAN code for procedure CAD. This code is a straightforward modification of that given in the CIRA publication. Displacement of the lower altitude boundary from 90 to 125 km, the neglect of hydrogen and argon, and the substitution of an analytical calculation for the numerical integration of the diffusion equation are the only changes made to the published CIRA code.

Section II of this report discusses the problem of the efficient calculation of the CIRA model, Section III addresses the problem of temperature-profile representation, Section IV addresses the computational changes needed to increase the lower boundary to 125 km, Section V examines the accuracy and efficiency of CAD in comparison with ADEN, and Section VI provides a summary.

II. CIRA 1972 MODEL

The CIRA 1972 atmospheric model attempts to deduce the thermal and compositional structure of the upper atmosphere from the observation of drag effects upon earth-orbiting satellites. This model formulates these structures in terms of a number of variable parameters, and the values of these parameters are adjusted to provide the closest approximation to the actually-observed drag values. While it is well-recognized that this solution is not unique as far as the thermal and compositional results are concerned (e.g., several different composition combinations can be derived to produce good matches with the drag data), the total mass density, being directly related to the measured drag, is represented rather well.

The CIRA assumes a fixed temperature and fixed densities for the atmospheric gas species at a lower boundary of 90 km. Up to 100 km a condition of mixing is assumed to exist. Above 100 km each constituent is assumed to be in separative diffusive equilibrium such that its number density n at any altitude z may be obtained by integrating the diffusion equation

$$d\ln n = -(1+\alpha)d\ln T - (mg/kT)dz \tag{1}$$

where α is the thermal-diffusion coefficient, m is the species mass, g is the acceleration due to gravity, k is Boltzmann's constant, and T is temperature. The integration of Equation 1 requires evaluation of the integral

$$I = \int \frac{mg}{kT}\, dz \tag{2}$$

where the formula

$$g = 9.80665/(1 + z/R)^2 \tag{3}$$

with R = 6356.766 km being the radius of the earth, is used to represent the variation of gravitational acceleration with altitude. The temperature profile formulas used by the CIRA, however, do not yield an integral which can be evaluated analytically. This is the primary task of the present study: to find an alternate representation for the temperature profile which remains a good approximation to the CIRA while yielding an analytically-integrable form in the diffusion equation.

III. TEMPERATURE PROFILE

In the altitude region above 125 km the CIRA uses the formula

$$T_z = T_{125} + \frac{2}{\pi}(T_\infty - T_{125}) \tan^{-1} \tag{4}$$

$$\{.0852718 \left[(T_{125}-183)/(T_\infty-T_{125})\right] \ (z-125) \left[1+.0000045(z-125)^{2.5}\right]\}$$

for the temperature profile. T_z, the temperature at altitude z, is asymptotic to an "exospheric" temperature T_∞ at high altitudes. T_{125} is given by this model in terms of T_∞

$$T_{125} = 371.6678 + 0.0518806\ T_\infty$$

$$- 294.3505 \exp(-0.00216222\ T_\infty) \tag{5}$$

The form of T_z in Equation 4 does not yield an analytically-integrable expression in Equation 2. It has been shown (Ref. 3,4) that a form which does yield an analytically-integrable expression is

$$T_z = T_\infty - (T_\infty - T_{125}) \exp(-sz') \tag{6}$$

where s is a constant "shape parameter" and

$$z' = (z-125)\ (R+125)/(R+z) \tag{7}$$

is the geopotential height above z = 125 km.

As we are interested here in altitudes small with respect to the earth's radius, this equation closely approximates a simple exponential-like asymptotic approach to T_∞ at high altitudes. The more complex exponential argument is required for exact integrability when the altitude variation of g (Equation 3) is taken into account. If Equation 6 is adopted for the temperature profile, Equation 2 may be analytically integrated to yield as a solution to Equation 1 for gas species i

$$n_i(z) = n_i(125) \left(\frac{1-a}{1-ae^{-sz'}}\right)^{1+\alpha+1/sH_i} e^{-z'/H_i} \tag{8}$$

$$a = (T_\infty - T_{125})/T_\infty$$

$$H_i = kT_\infty/m_i g_{125}$$

We thus find that it will be possible to replace the numerical integration of the diffusion equation as employed by the CIRA code ADEN with a simple closed, analytic expression (Equation 8) if Equation 6 is capable of fitting the CIRA temperature profiles sufficiently accurately. That this is possible has been verified by fitting these CIRA temperature profiles (using values at 25 km intervals from 125 to 500 km altitude) by Equation 6 for T_∞ = 500 to 1900 K in 100-K steps, requiring T_∞ and T_{125} to be the same for the two profile forms and finding the value of s for each value of T_∞ to give the closest agreement between the two profiles. The results of this exercise are shown in Figure 1. Here are shown the profile results, with the continuous curves representing the CIRA model and the discrete points representing the fits. From measurements of exospheric temperature at Millstone Hill by the incoherent scatter radar method (Ref. 5) it has been shown that the random (unmodelable) variability of T_∞ is on the order of 50 K. Thus the agreements shown in Figure 1 are considered acceptable in light of the uncertainty in our knowledge of the instantaneous state of the upper atmosphere.

Figure 2 plots s vs T_∞ and fits this trend by the equation

$$s = a_1 + a_2/(T_\infty + a_3) \qquad \mathrm{km}^{-1} \tag{9}$$

with the results shown in Table 1 for a_1, a_2, and a_3. The continuous curve in this figure represents the s values determined to fit the CIRA profiles while the data points represent the fit of Equation 9 to these values.



Fig. 1. CIRA temperature profiles (solid lines) and the fits afforded by Equation 6 (symbols).



Fig. 2. Variation of s with exospheric temperature for CIRA model (solid line) and the fit afforded by Equation 9 (symbols).

TABLE I

FITTED s AND BOUNDARY-CONCENTRATION PARAMETERS (EQUATION 10)

	a_1	a_2	a_3
s	.0057	17.6	244
N_2	17.419	-110	133
O_2	16.583	-155	152
O	16.944	140	3163
H_e	13.397	97	335

IV. BOUNDARY CONDITIONS

The CIRA incorporates fixed, non-varying concentrations for the gas species at a lower boundary of 90 km. Because the temperature profile in the 90-125 km region varies according to the solar/geophysical conditions, however, we cannot assume fixed, non-varying concentrations for the gas species at the 125-km boundary adopted for the present purposes. Fortunately, the entire temperature profile in the CIRA is defined once T_∞ has been chosen such that it is possible to determine the concentrations of the gas species at 125 km altitude solely as a function of T_∞. Once these "boundary variations" are obtained, the region below 125 km may be abandoned.

The 125-km boundary variations for the four CIRA gas species under consideration are shown in Figure 3 as a function of T_∞ (continuous curves). For each species these boundary "data" have been fitted by the same formula (Equation 9) used to model the s variation. The data points in Figure 3 represent the fits to the model variations. Table 1 also lists the results for a_1, a_2, and a_3 for each gas species.



Fig. 3. Variations of the logarithms (base 10) of the 125-km concentrations versus exospheric temperature for the CIRA model (solid lines) and the fits afforded by Equation 9 (symbols). Each variation is normalized relative to its value at 500 K.

V. ACCURACY AND EFFICIENCY

The accuracy of CAD relative to the official CIRA code ADEN has been checked by comparing the outputs of the two routines for altitudes from 125 to 500 km (25-km steps) and solar-flux values from 70 to 280 units (1 unit = $10^{-22} Wm^{-2} Hz^{-1}$) (10-unit steps) with all other inputs set to zero. For no combination of these height-flux variable did CAD differ by more than 7% from ADEN. Below a flux value of 250 units, a 4% error was the maximum encountered. The average difference between CAD and ADEN was about 2%. This is relatively insignificant with respect to the uncertainties in our knowledge of the state of the atmosphere.

The execution of one call to CAD required 4 to 5 ms on a Harris 125 machine while ADEN required 137 to 140 ms. Thus CAD is some 30 times faster than the CIRA routine. ADEN requires essentially the same execution time for any altitude from 100 to 500 km. This is because the hydrogen concentration, though of little significance to total density below 500 km (it is not even included in the CIRA tables below this altitude), is included in the ADEN calculation to avoid any discontinuity at this altitude. Because the CIRA adopts a boundary variation for the hydrogen concentration at the 500-km level, the inclusion of hydrogen at lower altitudes requires the integration of the diffusion equation downward from 500 km to the altitude of interest. Thus the numerical integration of ADEN must cover the entire 100-to-500-km range no matter which altitude in this region is under consideration.

VI. SUMMARY

Procedure CAD is intended to be an efficient machine implementation of the CIRA 1972 atmospheric drag model, designed specifically for use in the approximation of atmospheric drag effects upon orbiting satellites. The code ADEN provided by the CIRA publication for this calculation is altered to (a) approximate the numerical integration of the diffusion equation by an analytic expression, (b) limit the altitude range of calculation to the satellite-orbit region in which significant drag effects are experienced, and (c) eliminate minor constituents of insignificant density contributions. These modifications render CAD some 30 times faster than ADEN while accuracy is maintained to within a few percent.

The appendix presents a FORTRAN implementation of procedure CAD, retaining the form of ADEN as much as possible. The input and output arguments for the two routines are identical.

REFERENCES

1. W. P. Seniw and R. Sridharan, "ANODE: An Analytic Orbit Determination System," Technical Note 1980-1, Lincoln Laboratory, M.I.T. (to be published).

2. COSPAR International Reference Atmosphere (Academie-Verlag, Berlin, 1972).

3. D. R. Bates, "Some Problems Concerning the Terrestrial Atmosphere Above the 100-km Level," Proc. Roy. Soc. London A-253, 451 (1959).

4. J. C. G. Walker, "Analytic Representation of Upper Atmospheric Densities Based on Jacchia's Static Diffusion Models," J. Atmos. Sci. 22, 462 (1965).

5. W. L. Oliver, "Millstone Hill Exospheric Temperature Models," Technical Note 1979-56, Lincoln Laboratory, M.I.T. (30 July 1979).

APPENDIX

Included in this appendix is the FORTRAN implementation of procedure CAD.

```
C     SUBROUTINE CAD (CALCULATE ATMOSPHERIC DESITY) IS A MODIFICATION OF
C     SUBROUTINE ADEN, WHICH IS THE FORTRAN PACKAGE INCLUDED IN THE CIRA (COSPAR
C     INTERNATIONAL REFERENCE ATMOSPHERE) 1972 PUBLICATION (ACADEMIE VERLAG,
C     BERLIN, 1972) FOR THE CALCULATION OF ATMOSPHERIC TEMPERATURE, COMPOSITION,
C     AND DENSITY ABOVE AN ALTITUDE OF 90 KILOMETERS.  THESE MODIFICATIONS ADAPT
C     THIS MODEL FOR THE SPECIFIC PURPOSE OF DENSITY CALCULATIONS IN THE SATELLIT
C     -ORBIT REGION OF SIGNIFICANT DRAG.  THREE MODIFICATIONS HAVE BEEN
C     IMPLEMENTED TO MAKE THE CALCULATIONS MORE EFFICIENT OR THIS PURPOSE:
C          (1) REPLACEMENT OF THE NUMERICAL INTEGRATION OF THE BAROMETRIC
C              EQUATION BY AN ANALYTIC APPROXIMATION
C          (2) DISPLACEMENT OF THE LOWER BOUNDARY TO 125 KILOMETERS
C          (3) ELIMINATION OF CONSTITUENTS OF NEGLIGIBLE DRAG CONTRIBUTION.
C
      SUBROUTINE CAD (AMJD,SUN,SAT,GEO,TEMP,AL1ON,AMMW,RHO)
      DIMENSION SUN(2),SAT(3),GEO(3),TEMP(2),AL1ON(6)
C
C     ARGUMENTS REQUIRED OF CALLING (SUB)PROGRAM
C      AMJD      DATE AND TIME, IN MODIFIED JULIAN DAYS AND FRACTION
C                THEREOF (MJD=JD-2400000.5)
C      SUN(1)    RIGHT ASCENSION OF THE SUN, IN RADIANS
C      SUN(2)    DECLINATION OF THE SUN, IN RADIANS
C      SAT(1)    RIGHT ASCENSION OF THE POINT IN QUESTION, IN RADIANS
C      SAT(2)    DECLINATION (GEOCENTRIC LATITUDE) OF THE POINT IN
C                QUESTION, IN RADIANS
C      SAT(3)    HEIGHT OF THE POINT IN QUESTION, IN KILOMETERS
C      GEO(1)    10.7 CM SOLAR FLUX, IN UNITS OF 1.0E-22*WATT*(M**-2)*
C                (HERTZ**-1), FOR A TABULAR TIME 1.71 DAYS EARLIER
C      GEO(2)    10.7 CM SOLAR FLUX, AVERAGED OVER FOUR SOLAR ROTATIONS,
C                CENTERED ON THE TIME IN QUESTION
C      GEO(3)    THE GEOMAGNETIC PLANETARY THREE-HOUR-RANGE INDEX
C                K-SUB-P (3- = 2.667, 30 = 3.000, 3+ = 3.333, ETC.),
C                FOR A TABULAR TIME 0.279 DAYS EARLIER
C
C     ARGUMENTS RETURNED TO CALLING (SUB)PROGRAM
C
C      TEMP(1)   EXOSPHERIC TEMPERATURE ABOVE THE POINT IN QUESTION,
C                KELVIN
C      TEMP(2)   TEMPERATURE AT THE POINT IN QUESTION, KELVIN
C      AL1ON(1)  COMMON LOGARITHM OF THE N2 NUMBER-DENSITY, IN M**-3
C      AL1ON(2)  COMMON LOGARITHM OF THE O2 NUMBER-DENSITY
C      AL1ON(3)  COMMON LOGARITHM OF THE O  NUMBER-DENSITY
C      AL1ON(5)  COMMON LOGARITHM OF THE HE NUMBER-DENSITY
C      AMMW      MEAN-MOLECULAR-WEIGHT
C      RHO       TOTAL MASS-DENSITY, IN KG*(M**-3)
C
```

```
      DIMENSION ALN(6),ALPHA(5),AMW(6),BV(5)
C
C      THE ALPHA ARE THE THERMAL DIFFUSION COEFFICIENTS IN EQUATION (6)
      DATA ALPHA(1) /0.0/
      DATA ALPHA(2) /0.0/
      DATA ALPHA(3) /0.0/
      DATA ALPHA(5) /-0.38/
C
C        AL10 IS ALOG(10.)
C
      DATA AL10    /2.3025851/
C
C      THE AMW ARE THE MOLECULAR WEIGHTS IN THE ORDER N2,O2,O,HE
C
      DATA AMW(1)  /28.0134/
      DATA AMW(2)  /31.9988/
      DATA AMW(3)  /15.9994/
      DATA AMW(5)  / 4.0026/
C
C      AVOGAD IS AVOGADROS NUMBER IN MKS UNITS
C
      DATA AVOGAD  /6.02257E26/
C
C      CONS25 IS SIN(PI/4.0)**3, USED IN EQUATION (25)
C
      DATA CONS25  /0.35355339/
C
C      FOURPI IS 4.0*PI
C
      DATA FOURPI  /12.566371/
C
C      PIOV4 IS PI/4.0
C
      DATA PIOV4   /0.78539816/
C
C              TWOPI IS 2.0*PI
C
      DATA TWOPI   /6.2831853/
C
C              CONCENTRATION BOUNDARY VARIATIONS AT 125 KILOMETERS ALTITUDE
C              (ALOG10 (M**-3))
C
      BVN2(TINF)=17.419-110./(TINF+ 133.)
      BVO2(TINF)=16.583-155./(TINF+ 152.)
      BVO (TINF)=16.944+140./(TINF+3163.)
```

```
      BVHE(TINF)=13.397+ 97./(TINF+ 335.)
C
C               S VARIATION (KM**-1)
C
      SS(TINF)=.0057+17.6/(TINF+244.)
C
C               EQUATION (14)
C
      TSUBC=379.0+3.24*GEO(2)+1.3*(GEO(1)-GEO(2))
C
C               EQUATION (15)
C
      ETA=0.5*ABS(SAT(2)-SUN(2))
      THETA=0.5*ABS(SAT(2)+SUN(2))
C
C               EQUATION (16)
C
      H=SAT(1)-SUN(1)
      TAU=H-0.64577182+0.10471976*SIN(H+0.75049158)
C
C               EQUATION (17)
C
      C=COS(ETA)**2.2
      S=SIN(THETA)**2.2
      DF=S+(C-S)*ABS(COS(0.5*TAU))**3
      TSUBL=TSUBC*(1.0+0.3*DF)
C
C               EQUATION (18)
C
      EXPKP=EXP(GEO(3))
      DTG18=28.0*GEO(3)+0.03*EXPKP
C
C               EQUATION (20)
C
      DTG20=14.0*GEO(3)+0.02*EXPKP
      DLR20=0.012*GEO(3)+1.2E-5*EXPKP
C
C      THE FOLLOWING STATEMENTS EFFECT A CONTINUOUS TRANSITION FROM
C      EQUATION (20) AT HEIGHTS WELL BELOW 350 KM TO EQUATION (18) AT
C      HEIGHTS WELL ABOVE 350 KM
C
      F=0.5*(TANH(0.04*(SAT(3)-350.0))+1.0)
      DLRGM=DLR20*(1.0-F)
      DTG=DTG20*(1.0-F)+DTG18*F
C
```

```
      TINF=TSUBL+DTG
      TEMP(1)=TINF
C
C                EQUATION (9)
C
      TSUBX=371.6678+0.0518806*TINF-294.3503*EXP(-0.00216222*TINF)
C
C                EVALUATE TEMPERATURE AT SAT(3)
C
      SV=SS(TINF)
      ZPRIME=(SAT(3)-125.)*6481.766/(6356.766+SAT(3))
      EXPSZP=EXP(-SV*ZPRIME)
      TEMP(2)=TINF-(TINF-TSUBX)*EXPSZP
C
C               EVALUATE SPECIES CONCENTRATIONS
C
      BV(1)=BVN2(TINF)
      BV(2)=BVO2(TINF)
      BV(3)=BVO (TINF)
      BV(5)=BVHE(TINF)
      A=(TINF-TSUBX)/TINF
      ARG=(1.-A)/(1.-A*EXPSZP)
      H=1.380622E-23*TINF/(1.660531E-27*9.4320573)/1000.
      DO 14 I=1,5
      IF (I.EQ.4) GO TO 14
      HI=H/AMW(I)
      ALN(I)=BV(I)*AL10+ALOG(ARG)*(1.+ALPHA(I)+1./(SV*HI))-ZPRIME/HI
   14 CONTINUE
C
C                EQUATION (23)
C
   11 CAPPHI=AMOD((AMJD-36204.0)/365.2422,1.0)
C
C                EQUATION (22)
C
      TAU=CAPPHI+0.09544*((0.5+0.5*SIN(TWOPI*CAPPHI+6.035))**1.650-0.5)
      GOFT=0.02835+0.3817*(1.0+0.4671*SIN(TWOPI*TAU+4.137))
     1*SIN(FOURPI*TAU+4.259)
      FOFZ=(5.876E-7*SAT(3)**2.331+0.06328)*EXP(-2.868E-3*SAT(3))
C
C                EQUATION (21)
C
      DLRSA=FOFZ*GOFT
C
C                EQUATION (24)
```

```
C
      DLRSL=0.014*(SAT(3)-90.0)*EXP(-0.0013*(SAT(3)-90.0)**2)*SIGN
     1(1.0,SAT(2))*SIN(TWOPI*CAPPHI+1.72)*SIN(SAT(2))**2
C
C       SUM THE DELTA-LOG-RHOS AND APPLY TO THE NUMBER-DENSITIES
C
      DLR=AL10*(DLRGM+DLRSA+DLRSL)
      DO 12 I=1,5
      IF (I.EQ.4) GO TO 12
      ALN(I)=ALN(I)+DLR
   12 CONTINUE
C
C                  EQUATION (25)
C
      DLNHE=0.65*ABS(SUN(2)/0.4091609)*(SIN(PIOV4-0.5*SAT(2)*SIGN
     1(1.0,SUN(2)))**3-CONS25)
      ALN(5)=ALN(5)+AL10*DLNHE
C
C       COMPUTE MASS-DENSITY AND MEAN-MOLECULAR-WEITHT AND CONVERT
C       NUMBER-DENSITY LOGARITHMS FROM NATURAL TO COMMON
C
      SUMN=0.0
      SUMNM=0.0
      DO 13 I=1,5
      IF (I.EQ.4) GO TO 13
      AN=EXP(ALN(I))
      SUMN=SUMN+AN
      SUMNM=SUMNM+AN*AMW(I)
      AL10N(I)=ALN(I)/AL10
   13 CONTINUE
      AMMW=SUMNM/SUMN
      RHO=SUMNM/AVOGAD
      RETURN
      END
```

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: ESD-TR-80-28	2. GOVT ACCESSION NO.: AD-A087 447
3. RECIPIENT'S CATALOG NUMBER	
4. TITLE (and Subtitle): CAD: A Code for the Calculation of Atmospheric Density	5. TYPE OF REPORT & PERIOD COVERED: Technical Note
	6. PERFORMING ORG. REPORT NUMBER: Technical Note 1980-20
7. AUTHOR(s): William L. Oliver	8. CONTRACT OR GRANT NUMBER(s): F19628-80-C-0002
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Lincoln Laboratory, M.I.T., P.O. Box 73, Lexington, MA 02173	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: Program Element Nos. 12424F, 63428F; Project No. 2129
11. CONTROLLING OFFICE NAME AND ADDRESS: Air Force Systems Command, USAF, Andrews AFB, Washington, DC 20331	12. REPORT DATE: 2 April 1980
	13. NUMBER OF PAGES: 28
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office): Electronic Systems Division, Hanscom AFB, Bedford, MA 01731	15. SECURITY CLASS. (of this report): Unclassified
	15a. DECLASSIFICATION DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

None

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

atmospheric density satellite drag

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

An efficient machine implementation of the 1972 COSPAR International Reference Atmosphere (CIRA) is developed for use in evaluating atmospheric-drag effects upon earth-orbiting satellites. Restriction of this procedure to the satellite-orbit region of significant drag and the use of an alternate temperature profile representation have enabled a factor of 30 improvement in machine execution time compared with that required by the code supplied by CIRA publication while incurring only a few percent difference in the density results.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

207650